Mathematics in Lean

Release 0.1

Jeremy Avigad Kevin Buzzard Robert Y. Lewis Patrick Massot

CONTENTS

| 1 Introduction | | | | | |
|----------------|-------|--|--|--|--|
| | 1.1 | Getting Started | | | |
| | 1.2 | Overview | | | |
| 2 | Basic | c Skills | | | |
| | | Calculating | | | |
| | 2.2 | Proving Identities in Algebraic Structures | | | |
| | | Using Theorems and Lemmas | | | |

CHAPTER

ONE

INTRODUCTION

1.1 Getting Started

So, you are ready to formalize some mathematics. Maybe you have heard that formalization is the future (say, from the article, The Mechanization of Mathematics or the talk, The Future of Mathematics), and you want in. Or maybe you have played the Natural Number Game and you are hooked. Maybe you have heard about Lean and its library mathlib through online chatter and you want to know what the fuss is about. Or maybe you like mathematics and you like computers, are you have some time to spare. If you are in any of these situations, this book is for you.

Although you can read a pdf or html version of this book online, it designed to be read interactively, running Lean from inside the VS Code editor. To get started:

- 1. Install Lean, VS Code, and mathlib following the instructions in the mathlib repository.
- 2. In a terminal, type leanproject get mathematics_in_lean to set up a working directory for this tutorial.
- 3. Type code mathematics_in_lean to open that directory in VS Code.

Opening the file welcome.lean will simultaneously open this tutorial in a VS Code window. To update to a newer version of the tutorial, type git pull && leanproject get-mathlib-cache inside the mathematics_in_lean folder.

Every once in a while, you will see code snippet like this:

```
#eval "Hello, World!"
```

Clicking on the try it! button in the upper right corner will open a copy of the snippet in an examples folder and open it in a window so that you can experiment with it. This book provides lots of challenging exercises for you to do that way.

1.2 Overview

Put simply, Lean is a tool for building complex expressions in a formal language known as *dependent type theory*. Every expression has a *type*. Some expressions have types like \mathbb{N} or $\mathbb{N} \to \mathbb{N}$. These are mathematical objects.

```
#check 2 + 2
def f (x : \mathbb{N}) := x + 3
#check f
```

Some expressions have type *Prop*. These are mathematical statements.

```
#check 2 + 2 = 4

def fermat_last_theorem :=
\forall x y z n : \mathbb{N}, n > 2 \rightarrow x * y * z \neq 0 \rightarrow x^n + y^n \neq z^n
#check fermat_last_theorem
```

Some expressions have a type, P, where P itself has type Prop. Such an expression is a proof of the proposition P.

```
theorem easy : 2 + 2 = 4 := rfl

#check easy
theorem hard : fermat_last_theorem := sorry
#check hard
```

If you manage to construct an expression of type *fermat_last_theorem* and Lean accepts it as a term of that type, you have done something very impressive. (Using sorry is cheating, and Lean knows it.) So now you know the game. All that is left to learn are the rules.

This book is complementary to a companion tutorial, Theorem Proving in Lean, which provides a more thorough introduction to the underlying logical framework and core syntax of Lean. *Theorem Proving in Lean* is for people who prefer to read a user manual cover to cover before using a new dishwasher. If you are the kind of person who prefers to hit the *start* button and figure out how to activate the potscrubber feature later, it makes more sense to start here and refer back to *Theorem Proving in Lean* as necessary.

Another thing that distinguishes *Mathematics in Lean* from *Theorem Proving in Lean* is that here we place a much greater emphasis on the use of *tactics*. Given that were are trying to build complex expressions, Lean offers two ways of going about it: we can write down the expressions themselves (that is, suitable text descriptions thereof), or we can provide Lean with *instructions* as to how to construct them. For example, the following expression represents a proof of the fact that if n is even then so is m * n:

```
import data.nat.parity
open nat

example : ∀ m n, even n → even (m * n) :=
assume m n ⟨k, (hk : n = 2 * k)⟩,
have hmn : m * n = 2 * (m * k),
by rw [hk, mul_left_comm],
show ∃ 1, m * n = 2 * 1,
from ⟨_, hmn⟩
```

The *proof term* can be compressed to a single line:

```
example : \forall m n, even n \rightarrow even (m * n) := \lambda m n \langlek, hk\rangle, \langlem * k, by rw [hk, mul_left_comm]\rangle
```

The following is, instead, a *tactic-style* proof of the same theorem:

```
import data.nat.parity tactic
open nat

example : ∀ m n, even n → even (m * n) :=
begin
    rintros m n ⟨k, hk⟩,
    use m * k,
```

```
rw [hk, mul_left_comm]
end
```

As you enter each line of such a proof in VS Code, Lean displays the *proof state* in a separate window, telling you what facts you have already established and what tasks remain to prove your theorem. You can replay the proof by stepping through the lines, since Lean will continue to show you the state of the proof at the point where the cursor is. In this example, you will then see that the first line of the proof introduces m and n (we could have renamed them at that point, if we wanted to), and also decomposes the hypothesis even n to a k and the assumption that m = 2 * k. The second line, use m * k, declares that we are going to show that m * n is even by showing m * n = 2 * (m * k). The last line uses the rewrite tactic to replace n by 2 * k in the goal and then swap the m and the 2 to show that the two sides of the equality are the same.

The ability to build a proof in small steps with incremental feedback is extremely powerful. For that reason, tactic proofs are often easier and quicker to write than proof terms. There isn't a sharp distinction between the two: tactic proofs can be inserted in proof terms, as we did with the phrase by rw [hk, mul_left_comm] in the example above. We will also see that, conversely, it is often useful to insert a short proof term in the middle of a tactic proof. That said, in this book, our emphasis will be on the use of tactics.

In our example, the tactic proof can also be reduced to a one-liner:

```
example : \forall m n, even n \rightarrow even (m * n) := by rintros m n \langlek, hk\rangle; use m * k; rw [hk, mul_left_comm]
```

Here were have used tactics to carry out small proof steps. But they can also provide substantial automation, and justify longer calculations and bigger inferential steps. For example, we can invoke Lean's simplifier with specific rules for simplifying statements about parity to prove our theorem automatically.

```
example : \forall m n, even n \rightarrow even (m * n) := by intros; simp * with parity_simps
```

Another big difference between the two introductions is that *Theorem Proving in Lean* depends only on core Lean and its built-in tactics, whereas *Mathematics in Lean* is built on top of Lean's powerful and ever-growing library, *mathlib*. As a result, we can show you how to use some of the mathematical objects and theorems in the library, and some of the very useful tactics. This book is not meant to be used as an overview of the library; the mathlib web pages contain extensive documentation. Rather, our goal is to introduce you to the style of thinking that underlies that formalization, so that you are comfortable browsing the library and finding things on your own.

Interactive theorem proving can be frustrating, and the learning curve is steep. But the *Lean* community is very welcoming to newcomers, and people are available on the Lean Zulip chat group round the clock to answer questions. We hope to see you there, and have no doubt that soon enough you, too, will be able to answer such questions and contribute to the development of *mathlib*.

So here is your mission, should you choose to accept it: dive in, try the exercises, come to Zulip with questions, and have fun. But be forewarned: interactive theorem prover will challenge you to think about mathematics and mathematical reasoning in fundamentally new ways. Your life may never be the same.

Acknowledgment. We are grateful to Gabriel Ebner for setting up the infrastructure for running this tutorial in VS Code.

1.2. Overview 3

CHAPTER

TWO

BASIC SKILLS

This chapter is designed to introduce you to the nuts and bolts of mathematical reasoning in Lean: calculating, applying lemmas and theorems, and carrying out proof by induction.

2.1 Calculating

We generally learn to carry out mathematical calculations without thinking of them as proofs. But when we justify each step in a calculation, as Lean requires us to do, the net result is a proof that the left-hand side of the calculation is equal to the right-hand side.

In Lean, stating a theorem is tantamount to stating a goal, namely, the goal of proving the theorem. Lean provides the rewrite tactic, abbreviated rw, to replace the left-hand side of an identity by the right-hand side in the goal. If a, b, and c are real numbers, mul_assoc a b c is the identity a * b * c = a * (b * c) and mul_assoc a b is the identity a * b = b * a. In Lean, multiplication associates to the left, so the left-hand side of mul_assoc could also be written (a * b) * c. However, it is generally good style to be mindful of Lean's notational conventions and leave out parentheses when Lean does as well.

Let's try out rw.

```
import data.real.basic

example (a b c : R) : (a * b) * c = b * (a * c) :=
begin
    rw mul_comm a b,
    rw mul_assoc b a c
end
```

As you move your cursor past each step of the proof, you can see the goal of the proof change. The import line at the beginning of the example imports the theory of the real numbers from mathlib. For the sake of brevity, we generally suppress information like this when it is repeated from example to example. Clicking the try me! button displays all the example as it is meant to be processed and checked by Lean.

Incidentally, you can type the \mathbb{R} character as \R or $\$ in the VS Code editor. The symbol doesn't appear until you hit space or the tab key. If you hover over a symbol when reading a Lean file, VS Code will show you the syntax that can be used to enter it. If your keyboard does not have a backslash, you can change the leading character by changing the lean.input.leader setting in VS Code.

Try proving these identities, in each case replacing sorry by a tactic proof. With the rw tactic, you can use a left arrow to reverse an identity. For example, rw ← mul_assoc a b c replaces a * (b * c) by a * b * c in the current goal.

```
example (a b c : R) : (c * b) * a = b * (a * c) :=
begin
    sorry
end

example (a b c : R) : a * (b * c) = b * (a * c) :=
begin
    sorry
end
```

You can also use identities like mul_assoc and mul_comm without arguments. In this case, the rewrite tactic tries to match the left-hand side with an expression in the goal, using the first pattern it finds.

```
example (a b c : R) : a * b * c = b * c * a :=
begin
  rw mul_assoc,
  rw mul_comm
end
```

You can also provide *partial* information. For example, mul_comm a matches any pattern of the form a *? and rewrites it to? * a. Try doing the first of these examples without providing any arguments at all, and the second with only one argument.

```
example (a b c : R) : a * (b * c) = b * (c * a) :=
begin
    sorry
end

example (a b c : R) : a * (b * c) = b * (a * c) :=
begin
    sorry
end
```

In the Lean editor mode, when a cursor is in the middle of a tactic proof, Lean reports on the current *proof state*. A typical proof state in Lean might look as follows:

```
1 goal x y : \mathbb{N}, h_1 : prime x, h_2 : \neg even x, h_3 : y > x \vdash y \ge 4
```

The lines before the one that begins with \vdash denote the *context*: they are the objects and assumptions currently at play. In this example, these include two objects, x and y, each a natural number. They also include three assumptions, labelled h_1 , h_2 , and h_3 . In Lean, everything in a context is labelled with an identifier. You can type these subscripted labels as $h \ 1$, $h \ 2$, and $h \ 3$, but any legal identifiers would do: you can use $h \ 1$, $h \ 2$, $h \ 3$ instead, or $f \ 0$, $b \ ar$, and $b \ az$. The last line represents the *goal*, that is, the fact to be proved. Sometimes people use *target* for the fact to be proved, and *goal* for the combination of the context and the target. In practice, the intended meaning is usually clear.

You an also use rw with facts from the local context.

```
example (a b c d e f : R) (h : a * b = c * d) (h' : e = f) :
    a * (b * e) = c * (d * f) :=
begin
    rw h',
    rw ←mul_assoc,
```

```
rw h,
rw mul_assoc
end
```

Try these:

```
example (a b c d e f : R) (h : b * c = e * f) :
    a * b * c * d = a * e * f * d :=
begin
    sorry
end

example (a b c d : R) (hyp : c = b * a - d) (hyp' : d = a * b) : c = 0 :=
begin
    sorry
end
```

For the second one, you can use the theorem sub_self , where sub_self a is the identity a - a = 0.

We now introduce some useful features of Lean. First, multiple rewrite commands can be carried out with a single command, by listing the relevant identities within square brackets. Second, when a tactic proof is just a single command, we can replace the begin ... end block with a by.

```
example (a b c d e f : R) (h : a * b = c * d) (h' : e = f) :
    a * (b * e) = c * (d * f) :=
begin
    rw [h', ←mul_assoc, h, mul_assoc]
end

example (a b c d e f : R) (h : a * b = c * d) (h' : e = f) :
    a * (b * e) = c * (d * f) :=
by rw [h', ←mul_assoc, h, mul_assoc]
```

You still see the incremental progress by placing the cursor after a comma in any list of rewrites.

Another trick is that we can declare variables once and forall outside an example or theorem. When Lean sees them mentioned in the statement of the theorem, it includes them automatically.

```
variables a b c d e f : R

example (h : a * b = c * d) (h' : e = f) :
    a * (b * e) = c * (d * f) :=
by rw [h', ←mul_assoc, h, mul_assoc]
```

We can delimit the scope of the declaration by putting it in a section ... end block. Finally, Lean provides us with a means of checking and expression and determining its type:

```
section
variables a b c : R

#check a
#check a + b
#check (a : R)
#check mul_comm a b
#check (mul_comm a b : a * b = b * a)
#check mul_assoc c a b
#check mul_comm a
```

(continues on next page)

2.1. Calculating 7

```
#check mul_comm
#check @mul_comm
end
```

The #check command works for both objects and facts. In response to the command #check a, Lean reports that a has type \mathbb{R} . In response to the command #check mul_comm a b, Lean reports that mul_comm a b is a proof of the fact a + b = b + a. The command #check (a : \mathbb{R}) states our expectation that the type of a is \mathbb{R} , and Lean will raise an error if that is not the case. We will explain the output of the last three #check command later, but in the meanwhile, you can take a look at them, and experiment with some #check commands of your own.

Let's try some more examples. The theorem two_mul a says that a + a = 2 * a. The theorems add_mul and mul_add express the distributivity of multiplication over addition, and the theorem add_assoc expresses the associativity of addition. Use the #check command to see the precise statements.

```
example : (a + b) * (a + b) = a * a + 2 * (a * b) + b * b :=
begin
  rw [mul_add, add_mul, add_mul],
  rw [*add_assoc, add_assoc (a * a)],
  rw [mul_comm b a, *two_mul]
end
```

Whereas it is possible to figure out what it going on in this proof by stepping through it in the editor, it is hard to read on its own. Lean provides a more structured way of writing proofs like this using the calc keyword.

```
example : (a + b) * (a + b) = a * a + 2 * (a * b) + b * b :=
calc

(a + b) * (a + b)

= a * a + b * a + (a * b + b * b) :
    by rw [mul_add, add_mul, add_mul]
... = a * a + (b * a + a * b) + b * b :
    by rw [+add_assoc, add_assoc (a * a)]
... = a * a + 2 * (a * b) + b * b
    by rw [mul_comm b a, +two_mul]
```

Notice that there is no more begin ... end block: an expression that begins with calc is a *proof term*. A calc expression can also be used inside a tactic proof, but Lean interprets it as the instruction to use the resulting proof term to solve the goal exactly.

The calc syntax is finicky: the dots and colons and justification have to be in the format indicated above. Lean ignores whitespace like spaces, tabs, and returns, so you have some flexibility to make the calculation look more attractive. One way to write a calc proof is to outline it first using the sorry tactic for justification, make sure Lean accepts the expression modulo these, and then justify the individual steps using tactics.

Try proving the following identity using both a pure rw proof and a more structured calc proof:

```
example: (a + b) * (c + d) = a * c + a * d + b * c + b * d := sorry
```

The following exercise is a little more challenging. You can use the theorems listed underneath.

```
example (a b : R) : (a + b) * (a - b) = a^2 - b^2 :=
begin
    sorry
end

#check pow_two a
#check mul_sub a b c
#check add_mul a b c
#check add_sub a b c
#check sub_sub a b c
#check add_zero a
```

We can also perform rewriting in an assumption in the context. For example, rw mul_comm a b at hyp replaces a*b by b*a in the assumption hyp.

```
example (a b c d : R) (hyp : c = d * a + b) (hyp' : b = a * d) :
    c = 2 * a * d :=
begin
    rw hyp' at hyp,
    rw mul_comm d a at hyp,
    rw ← two_mul (a*d) at hyp,
    rw ← mul_assoc 2 a d at hyp,
    exact hyp
end
```

In the last step, the exact tactic can use hyp to solve the goal because at that point hyp matches the goal exactly.

We close this section by noting that mathlib provides a useful bit of automation with a ring tactic, which is designed to prove identities in any ring.

```
example : (c * b) * a = b * (a * c) :=
by ring

example : (a + b) * (a + b) = a * a + 2 * (a * b) + b * b :=
by ring

example : (a + b) * (a - b) = a^2 - b^2 :=
by ring

example (a b c d : R) (hyp : c = d * a + b) (hyp' : b = a * d) :
    c = 2 * a * d :=
begin
    rw [hyp, hyp'],
    ring
end
```

The ring tactic is imported indirectly when we import data.real.basic, but we will see in the next section that it can be used for calculations on structures other than the real numbers. It can be imported explicitly with the command import tactic.

2.1. Calculating 9

2.2 Proving Identities in Algebraic Structures

Mathematically, a ring consists of a set, R, operations $+\times$, and constants 0 and 1, and an operation $x\mapsto -x$ such that:

- R with + is an abelian group, with 0 as the additive identity and negation as inverse.
- Multiplication is associative with identity 1, and multiplication distributes over addition.

In Lean, with base our algebraic structures on *types* rather than sets. Modulo this difference, we can take the ring axioms to be as follows:

```
variables (R : Type*) [comm_ring R]

#check (add_assoc : V a b c : R, a + b + c = a + (b + c))
#check (add_comm : V a b : R, a + b = b + a)
#check (zero_add : V a : R, 0 + a = a)
#check (add_left_neg : V a : R, -a + a = 0)
#check (mul_assoc : V a b c : R, a * b * c = a * (b * c))
#check (mul_one : V a : R, a * 1 = a)
#check (one_mul : V a : R, 1 * a = a)
#check (mul_add : V a b c : R, a * (b + c) = a * b + a * c)
#check (add_mul : V a b c : R, (a + b) * c = a * c + b * c)
```

You will learn more about the square brackets in the first line later, but for the time being, suffice it to say that the declaration gives us a type, R, and a ring structure on R. Lean then allows us to use generic ring notation with elements of R, and to make use of a library of theorems about rings.

The names of some of the theorems should look familiar: they are exactly the ones we used to calculate with the real numbers in the last section. Lean is good not only for proving things about concrete mathematical structures like the natural numbers and the integers, but also for proving things about abstract structures, characterized axiomatically, like rings. Moreover, Lean supports *generic reasoning* about both abstract and concrete structures, and can be trained to recognized appropriate instances. So any theorem about rings can be applied to concrete rings like the integers, \mathbb{Z} , the rational numbers, \mathbb{Q} , and the complex numbers \mathbb{C} . It can also be applied to any instance of an abstract structure that extends rings, such as any *ordered ring* or any *field*.

Not all important properties of the real numbers hold in an arbitrary ring, however. For example, multiplication on the real numbers is commutative, but that does not hold in general. If you have taken a course in linear algebra, you will recognize that, for every n, the n by n matrices of real numbers for a ring in which commutativity fails. If we declare \mathbb{R} to be a *commutative* ring, in fact, all the theorems in the last section continue to hold when we replace \mathbb{R} by \mathbb{R} .

```
import tactic

variables (R : Type*) [comm_ring R]
variables a b c d : R

example : (c * b) * a = b * (a * c) :=
by ring

example : (a + b) * (a + b) = a * a + 2 * (a * b) + b * b :=
by ring

example : (a + b) * (a - b) = a^2 - b^2 :=
by ring

example (hyp : c = d * a + b) (hyp' : b = a * d) :
    c = 2 * a * d :=
begin
    rw [hyp, hyp'],
```

```
ring end
```

We leave it to you to check that all the other proofs go through unchanged.

The goal of this section is to strengthen the skills you have developed in the last section and apply them to reasoning axiomatically about rings. We will start with the axioms listed above, and use them to derive other facts. Most of the facts we prove are already in mathlib. We will give the versions we prove the same names to help you learn the contents of the library as well as the naming conventions. To avoid error messages from Lean, we will put our versions in a new namespace called my_ring.

The next example shows that we do not need add_zero or add_right_neg as ring axioms, because they follow from the other axioms.

```
namespace my_ring

variables {R : Type*} [ring R]

theorem add_zero (a : R) : a + 0 = a :=
by rw [add_comm, zero_add]

theorem add_right_neg (a : R) : a + -a = 0 :=
by rw [add_comm, add_left_neg]

end my_ring

#check @my_ring.add_zero
#check @add_zero
```

The net effect is that we can temporarily reprove a theorem in the library, and then go on using the library version after that. But don't cheat! In the exercises that follow, take care to use only the general facts about rings that we have proved earlier in this section.

(If you are paying careful attention, you may have noticed that we changed the round brackets in (R: Type*) for curly brackets in {R: Type*}. This declares R to be an *implicit argument*. We will explain what this means in a moment, but don't worry about it in the meanwhile.)

Here is a useful theorem:

```
theorem neg_add_cancel_left (a b : R) : -a + (a + b) = b :=
by rw [←add_assoc, add_left_neg, zero_add]
```

Prove the companion version:

```
theorem neg_add_cancel_right (a b : R) : (a + b) + -b = 0 :=
sorry
```

Use these to prove the following:

```
theorem add_left_cancel {a b c : R} (h : a + b = a + c) : b = c :=
sorry
theorem add_right_cancel {a b c : R} (h : a + b = c + b) : a = c :=
sorry
```

If you are clever, you can do each of them with three rewrites.

To illustrate, let's show that a * 0 = 0 follows from the ring axioms.

```
theorem mul_zero (a : R) : a * 0 = 0 :=
begin
  have h : a * 0 + a * 0 = a * 0 + 0,
  { rw [+mul_add, add_zero, add_zero] },
  rw add_left_cancel h
end
```

We have used a new trick! If you step through the proof, you can see what is going on. The have tactic introduces a new goal, a * 0 + a * 0 = a * 0 + 0, with the same context as the original goal. In the next line, we could have omitted the curly brackets, which serve as an inner begin ... end pair. Using them promotes a modular style of proof: Here the curly brackets could be omitted, the part of the proof inside the brackets establishes the goal that was introduced by the have. After that, we are back to proving the original goal, except a new hypothesis h has been added: having proved it, we are now free to use it. At this point, the goal is exactly the result of add_left_cancel h. We could equally well have closed the proof with apply add_left_cancel h or exact add_left_cancel h. We will discuss apply and exact in the next section.

Remember that multiplication is not assumed to be commutative, so the following theorem also requires some work.

```
theorem zero_mul (a : R) : 0 * a = 0 := sorry
```

By now, you should also be able replace each sorry in the next exercise with a proof, still using only facts about rings that we have established in this section.

```
theorem neg_eq_of_add_eq_zero {a b : R} (h : a + b = 0) : -a = b :=
sorry

theorem eq_neg_of_add_eq_zero {a b : R} (h : a + b = 0) : a = -b :=
sorry

theorem neg_zero : (-0 : R) = 0 :=
begin
    apply neg_eq_of_add_eq_zero,
    rw add_zero
end

theorem neg_neg (a : R) : -(-a) = a :=
sorry
```

We had to use the annotation (-0 : R) instead of 0 in the third theorem because without specifying R it is impossible for Lean to infer which 0 we have in mind.

In Lean, subtraction in a ring is defined to be addition of the additive inverse.

```
theorem sub_eq_add_neg (a b : R) : a - b = a + -b :=
rfl

example (a b : R) : a - b = a + -b :=
by reflexivity
```

The proof term rfl is short for reflexivity. Presenting it as a proof of a - b = a + -b forces Lean to unfold the definition and recognize both sides as being the same. The reflexivity tactic, which can be abbreviated as refl, does the same. This is an instance of what is known as a *definitional equality* in Lean's underlying logic. This means that not only can one rewrite with $sub_eq_add_neg$ to replace a - b = a + -b, but in some contexts you can use the two sides of the equation interchangeably. For example, you now have enough information to prove the theorem self sub from the last section:

```
theorem self_sub (a : R) : a - a = 0 :=
sorry
```

Extra points if you do it two different ways: once using rw, and once using either apply or exact.

For another example of definitional equality, Lean knows that 1 + 1 = 2 holds in any ring. With a bit of cleverness, you can use that to prove the theorem two_mul from the last section:

```
lemma one_add_one_eq_two : 1 + 1 = (2 : R) :=
by refl
theorem two_mul (a : R) : 2 * a = a + a :=
sorry
```

We close this section by noting that some of the facts about addition and negation that we established above do not need the full strength of the ring axioms, or even commutativity of addition. The weaker notion of a *group* can be axiomatized as follows:

```
variables (A : Type*) [add_group A]

#check (add_assoc : V a b c : A, a + b + c = a + (b + c))
#check (zero_add : V a : A, 0 + a = a)
#check (add_left_neg : V a : A, -a + a = 0)
```

It is conventional to use additive notation when a group the operation is commutative, and multiplicative notation otherwise. So Lean defines a multiplicative version as well as the additive version (and also their abelian variants, add_comm_group and comm_group).

```
wariables (G : Type*) [group G]

#check (mul_assoc : ∀ a b c : G, a * b * c = a * (b * c))
#check (one_mul : ∀ a : G, 1 * a = a)
#check (mul_left_inv : ∀ a : G, a-1 * a = 1)
```

If you are feeling cocky, try proving the following facts about groups, using only these axioms. You will need to prove a number of helper lemmas along the way. The proofs we have carried out in this section provide some hints.

```
variables {G : Type*} [group G]

#check (mul_assoc : ∀ a b c : G, a * b * c = a * (b * c))
#check (one_mul : ∀ a : G, 1 * a = a)
#check (mul_left_inv : ∀ a : G, a<sup>-1</sup> * a = 1)
```

```
namespace my_group

theorem one_mul (a : G) : 1 * a = a :=
sorry

theorem one_right_inv (a : G) : a * a<sup>-1</sup> = 1 :=
sorry

theorem mul_inv_rev (a b : G) : (a * b)<sup>-1</sup> = b<sup>-1</sup> * a <sup>-1</sup> :=
sorry

end my_group
```

2.3 Using Theorems and Lemmas

Rewriting is great for proving equations, but what about other sorts of theorems? For example, how can we prove an inequality, like the fact that $a+e^b \le a+e^c$ holds whenever $b \le c$? We have already seen that theorems can be applied to arguments and hypotheses, and that the apply and exact tactics can be used to solve goals. In this section, we will make good use of these tools.

Consider the library theorems le_refl and le_trans:

```
import data.real.basic

variables a b c : \mathbb{R}

#check (le_refl : \forall a : \mathbb{R}, a \leq a)
#check (le_trans : a \leq b \rightarrow b \leq c \rightarrow a \leq c)
```

The library designers have set the arguments to le_trans implicit, so that Lean will *not* let you provide them explicitly. Rather, it expects to infer them from the context in which they are used. For example, when hypotheses $h: a \le b$ and $h': b \le c$ are in the context, all the following work:

```
variables a b c : \mathbb{R} variables (h : a \leq b) (h' : b \leq c)

#check (le_refl : \forall a : real, a \leq a)
#check (le_refl a : a \leq a)
#check (le_trans : a \leq b \rightarrow b \leq c \rightarrow a \leq c)
#check (le_trans h : b \leq c \rightarrow a \leq c)
#check (le_trans h h' : a \leq c)
```

The apply tactic takes a proof of a general statement or implication, tries to match the conclusion with the current goal, and leaves the hypotheses, if any, as new goals. If the given proof matches the goal exactly, you can use the exact tactic instead of apply. So, all of these work:

```
\begin{array}{l} \textbf{example} \ (x \ y \ z \ : \ \mathbb{R}) \ (h_0 \ : \ x \le y) \ (h_1 \ : \ y \le z) \ : \ x \le z \ := \\ \textbf{begin} \\ \text{apply le_trans,} \\ \{ \ \text{apply } h_0 \ \}, \\ \text{apply } h_1 \\ \textbf{end} \\ \textbf{example} \ (x \ y \ z \ : \ \mathbb{R}) \ (h_0 \ : \ x \le y) \ (h_1 \ : \ y \le z) \ : \ x \le z \ := \\ \end{array}
```

(continues on next page)

14

```
\begin{array}{l} \textbf{begin} \\ \text{apply le_trans h}_0, \\ \text{apply h}_1 \\ \textbf{end} \\ \\ \textbf{example} \ (\texttt{x} \ \texttt{y} \ \texttt{z} : \mathbb{R}) \ (\texttt{h}_0 : \texttt{x} \le \texttt{y}) \ (\texttt{h}_1 : \texttt{y} \le \texttt{z}) : \texttt{x} \le \texttt{z} := \\ \textbf{by} \ \text{exact le_trans h}_0 \ \texttt{h}_1 \\ \\ \textbf{example} \ (\texttt{x} \ \texttt{y} \ \texttt{z} : \mathbb{R}) \ (\texttt{h}_0 : \texttt{x} \le \texttt{y}) \ (\texttt{h}_1 : \texttt{y} \le \texttt{z}) : \texttt{x} \le \texttt{z} := \\ \textbf{le_trans h}_0 \ \texttt{h}_1 \\ \\ \textbf{example} \ (\texttt{x} : \mathbb{R}) : \texttt{x} \le \texttt{x} := \\ \textbf{by} \ \text{apply le_refl} \\ \\ \textbf{example} \ (\texttt{x} : \mathbb{R}) : \texttt{x} \le \texttt{x} := \\ \textbf{by} \ \text{exact le_refl x} \\ \\ \textbf{example} \ (\texttt{x} : \mathbb{R}) : \texttt{x} \le \texttt{x} := \\ \textbf{le_refl x} \\ \end{array}
```

In the first example, applying le_trans creates two goals, and we use the curly braces to enclose the proof of the first one. In the fourth example and in the last example, we avoid going into tactic mode entirely: $le_trans h_0 h_1$ and $le_refl x$ are the proof terms we need.

Here are a few more library theorems:

```
#check (le_refl : \forall a, a \leq a) #check (le_trans : a \leq b \rightarrow b \leq c \rightarrow a \leq c) #check (lt_of_le_of_lt : a \leq b \rightarrow b < c \rightarrow a < c) #check (lt_of_lt_of_le : a < b \rightarrow b \leq c \rightarrow a < c) #check (lt_trans : a < b \rightarrow b < c \rightarrow a < c)
```

Use them together with apply and exact to prove the following:

```
\textbf{example} (a b c d e : \mathbb{R}) (h_0 : a \leq b) (h_1 : b < c) (h_2 : c \leq d) (h_3 : d < e) : a < e := sorry
```

In fact, Lean has a tactic that does this sort of thing automatically:

The linarith tactic is designed to handle linear arithmetic.

```
example (h : 2 * a \leq 3 * b) (h' : 1 \leq a) (h'' : d = 2) : d + a \leq 5 * b := by linarith
```

In addition to equations and inequalities in the context, linarith will use additional inequalities that you pass as arguments.

```
example (h : 1 \le a) (h' : b \le c) : 2 + a + exp b \le 3 * a + exp c := by linarith [exp_le_exp.mpr h']
```

Here are some more theorems in the library that can be used to establish inequalities on the real numbers.

```
import analysis.special_functions.exp_log

open real

**tcheck** (exp_le_exp : exp a \le exp b \lefta a \le b)

#check** (exp_lt_exp : exp a \le exp b \le a \le b)

#check** (log_le_log : 0 \le a \righta 0 \le b \righta (log b \le a \le b))

#check** (log_le_log : 0 \le a \righta a \le b \righta (log b \le b))

#check** (log_lt_log : 0 \le a \righta a \le b \righta log b)

#check** (add_le_add : a \le b \righta c \le d \righta a + c \le b + d)

#check** (add_lt_add_of_le_of_lt : a \le b \righta c \le d \righta a + c \le b + d)

#check** (add_lt_add_of_le_of_lt : a \le b \righta c \le d \righta a + c \le b + d)

#check** (add_nonneg : 0 \le a \righta 0 \le b \righta 0 \le a + b)

#check** (add_pos : 0 \le a \righta 0 \le b \righta 0 \le a + b)

#check** (add_pos_of_pos_of_nonneg : 0 \le a \righta 0 \le b \righta 0 \le a + b)

#check** (exp_pos : \forall a, 0 \le exp a)
```

Some of the theorems, exp_le_exp, exp_lt_exp, and log_le_log use a *bi-implication*, which represents the phrase "if and only if." (You can type it in VS Code with \lr of \iff). We will discuss this connective in greater detail in the next chapter. Such a theorem can be used with rw to rewrite a goal to an equivalent one:

```
example (a b : \mathbb{R}) (h : a \leq b) : exp a \leq exp b := begin

rw exp_le_exp,

exact h

end
```

In this section, however, we will use that fact that if $h: A \leftrightarrow B$ is such an equivalence, then h mp establishes the forward direction, $A \to B$, and h mpr establishes the reverse direction, $B \to A$. Here, mp stands for "modus ponens" and mpr stands for modus ponens reverse. Thus the following proof works:

```
\begin{array}{l} \textbf{example} \ (h_0 \ : \ a \ \le \ b) \ (h_1 \ : \ c \ < \ d) \ : \ a \ + \ exp \ c \ + \ e \ < \ b \ + \ exp \ d \ + \ e := \\ \textbf{begin} \\ \text{apply add\_lt\_add\_of\_lt\_of\_le,} \\ \{ \ apply \ add\_lt\_add\_of\_le\_of\_lt \ h_0, \\ \text{apply exp\_lt\_exp.2 } h_1 \ \}, \\ \text{apply le\_refl} \\ \textbf{end} \end{array}
```

The first line, apply add_lt_add_of_lt_of_le, creates two goals, and once again we use the curly brackets to separate the proof of the first from the proof of the second.

Try the following examples on your own. The example in the middle shows you that the norm_num tactic can be used to solve concrete numeric goals.

```
 \begin{array}{l} \textbf{example} \ (h_0 \ : \ d \le e) \ : \ c \ + \ exp \ (a + d) \ \le \ c \ + \ exp \ (a + e) \ := \\ \textbf{begin} \\ \text{sorry} \\ \textbf{end} \\ \end{array}
```

```
\begin{array}{l} \textbf{example} : (0 : \mathbb{R}) < 1 := \\ \textbf{by} \ \text{norm\_num} \\ \\ \textbf{example} \ (h : a \leq b) : \log \ (1 + \exp a) \leq \log \ (1 + \exp b) := \\ \textbf{begin} \\ \textbf{have} \ h_0 : 0 < 1 + \exp a, \\ \{ \ \text{sorry} \ \}, \\ \textbf{have} \ h_1 : 0 < 1 + \exp b, \\ \{ \ \text{sorry} \ \}, \\ \text{apply} \ (\log_{-} \log \log h_0 \ h_1) . 2, \\ \text{sorry} \\ \textbf{end} \end{array}
```

From these examples, it should be clear that being able to find the library theorems you need constitutes an important part of formalization. There are a number of strategies you can use:

- You can browse mathlib in its Github repository.
- You can use the API documentation on the mathlib web pages.
- You can rely on mathlib naming conventions and tab completion in the editor to guess a theorem name. In Lean, a
 theorem named A_of_B_of_C establishes something of the form A from hypotheses of the form B and C, where
 A, B, and C approximate the way we might read the goals out loud. So a theorem establishing something like x +
 y ≤ ... will probably start with add_le. Typing add_le and hitting tab will give you some helpful choices.
- If you right-click on an existing theorem in VS Code, the editor will jump to the file where it is defined, and you can find similar theorems nearby.
- You can use the library_search tactic, which tries to find the relevant theorem in the library.

```
import data.real.basic
import tactic

example (a : R) : 0 \le a^2 :=
begin
    -- library_search,
    exact pow_two_nonneg a
end
```

To try out library_search in this example, delete the exact command and uncomment the previous line. Using these tricks, see if you can find what you need to do the next example:

```
example (h : a \leq b) : c - exp b \leq c - exp a := begin sorry end
```

Also, confirm that linarith can do it with a bit of help.

Here is another example of an inequality:

```
example : 2*a*b \le a^2 + b^2 :=
begin
  have h : 0 \le a^2 - 2*a*b + b^2,
  calc
     a^2 - 2*a*b + b^2 = (a - b)^2 : by ring
     ... \geq 0 : by apply pow_two_nonneg,
  calc
```

```
2*a*b : by ring : by ring : ... \le 2*a*b + (a^2 - 2*a*b + b^2) : add_le_add (le_refl_) h :.. = a^2 + b^2 : by ring end
```

Mathlib tends to put spaces around binary operations like * and ^, but in this example, the more compressed format increases readability. There are a number of things worth noticing in this example. First, an expression $s \ge t$ is definitionally equivalent to $t \le s$. In principle, this means one should be able to use them interchangeably. But some of Lean's automation does not recognize the equivalence, so mathlib tends to favor \le over \ge . Second, we have used the ring tactic extensively. It is a real timesaver! Finally, notice that in the second line of the second calc proof, instead of writing by exact add_le_add (le_refl _) h, we can simply write the proof term add_le_add (le_refl _) h.

In fact, the only cleverness in the proof above is figuring out the hypothesis h. Once we have it, the second calculation involves only linear arithmetic, and linarith can handle it:

```
example : 2*a*b \leq a^2 + b^2 :=
begin
  have h : 0 \leq a^2 - 2*a*b + b^2,
  calc
  a^2 - 2*a*b + b^2 = (a - b)^2 : by ring
  ... \geq 0
  linarith
end
example : 2*a*b \leq b^2 : by apply pow_two_nonneg,
```

How nice! We challenge you to use these ideas to prove the following theorem. You can use the theorem abs_le_of_le_of_neg_le.

```
example : abs (a*b) < (a^2 + b^2) / 2 :=
sorry
#check abs_le_of_le_of_neg_le</pre>
```

If you managed to solve this, congratulations! You are well on your way to becoming a master formalizer.